

Curs pràctic de Maple

Pràctica 1

1 Càlculs Numèrics

En aquesta secció utilitzareu Maple per a fer alguns càlculs numèrics estàndard. L'habilitat de Maple per a donar resultats exactes a més de les aproximacions numèriques us donarà més opcions a l'hora de resoldre problemes.

1.1 Realitzant càlculs aritmètics exactes amb Maple

Utilitzar Maple per a realitzar càlculs numèrics és molt fàcil. Només s'ha d'introduir l'expressió numèrica i acabar la línia amb un **punt i coma** (;). Prement [**Retorn**] executareu la línia i el resultat apareixerà en color blau en el centre de la pantalla.

Exemple 1.1 Introduïu el següent càlcul simple i premeu [**Retorn**].

```
> 2+4;  
> 12*34567890;
```

Cada línia vermella és “viva” i es pot modificar en qualsevol moment.

Canvieu el “4” de la línia anterior per un “8” i premeu [**Retorn**].

Noteu com el resultat de color blau s'actualitza automàticament per a mostrar el nou resultat.

Exemple 1.2 Pel nostre exemple següent calculem 134^{39} .

```
> 134^39;
```

A diferència de la vostra calculadora, Maple us donarà la resposta exacte d'aquest problema, les 83 xifres que té.

Exemple 1.3

Maple pot fer càlculs amb fraccions sense convertir-les en expressions decimals:

```
> 3/5 + 5/9 + 7/12;
```

Exemple 1.4

Per a introduir l'arrel quadrada d'un nombre utilitzeu `sqrt()` :

```
> sqrt(24);
```

Noteu que Maple ha simplificat $\sqrt{24}$ però que ha deixat el resultat en forma exacta. En el proper apartat aprendreu com obtenir una aproximació decimal per a aquest nombre.

Exemple 1.5

Maple té introduïdes totes les constants matemàtiques importants. Per a introduir π escriviu `Pi`.

Fixeu-vos que es necessita un asterisc `*` per a indicar la multiplicació.

```
> 4*(3+Pi);
```

Una altra vegada Maple fa el càlcul però deixa el resultat en una forma exacta.

Exemple 1.6

A diferència de la vostra calculadora, Maple dona la solució exacta quan s'aplica a les funcions trigonomètriques.

```
> sin(5*Pi/3);  
> sec(Pi/4);
```

Per a obtenir l'invers del sinus d'un nombre utilitzeu la funció `arcsin()`:

```
> arcsin(-1);
```

Si demaneu a Maple que calculi un valor *indeterminat* respondrà amb un missatge d'error:

```
> tan(Pi/2);
```

Exemple 1.7

Per a introduir la funció exponencial e^x en Maple escriviu: `exp(x)` .

```
> exp(x);
```

I per a obtenir el nombre e escriviu: `exp(1)` .

```
> exp(1);
```

Exemple 1.8

Per a introduir la funció valor absolut $|x|$ en Maple escriviu: `abs(x)`.

Noteu que Maple dona la solució correcta i exacta per a la tercera línia ja que: $e - \pi < 0$

```
> abs(x);  
> abs(-3);  
> abs(exp(1)-Pi);
```

Exemple 1.9

Maple té moltes comandes per a realitzar càlculs específics amb nombres. Les podreu aprendre així que es vagin necessitant. Aquí presentem un últim exemple per ara: si teniu un nombre enter i voleu obtenir la seva descomposició en factors primers podeu utilitzar la comanda de Maple `ifactor()`. Podeu practicar tot el que vulgueu canviant el nombre que hi ha aquí sota.

```
> ifactor(31722722304);
```

Exemple 1.10

Pot haver hi molts moments en els que vulgueu introduir més d'una comanda en la mateixa línia. Això es pot fer en Maple, només cal assegurar-se d'acabar **cada una** de les comandes amb un punt i coma (;). També ajudarà posar espais en blanc entre les comandes. Quan premeu **[Retorn]** totes les expressions s'executaran i els resultats apareixeran, en ordre, en un únic camp de resultats.

```
> sin(Pi/3);   cos(Pi/3);   tan(Pi/3);
```

Exemple 1.11

Per a calcular i mostrar una successió de nombres utilitzeu la comanda `seq(. .)`. Aquí calculem els quadrats dels 100 primers nombres naturals

```
> seq(k^2,k=1..100);
```

1.2 Aproximacions numèriques utilitzant la comanda `evalf()`

Recordeu que en l'apartat anterior hem demanat a Maple que fes la suma de tres fraccions i que ha donat com a resultat una altra fracció. Aquest tipus d'aritmètica exacta és molt útil però hi ha vegades que preferirem una resposta en forma d'expressió decimal. La comanda de Maple `evalf()` realitza aquesta tasca per nosaltres.

Exemple 1.1

Compareu els resultats de les dues línies següents.

```
> 3/5+5/9+7/12;
> evalf(3/5+5/9+7/12);
```

Exemple 1.2

Donar un nom al resultat d'un determinat càlcul fa que sigui més fàcil utilitzar-lo en un càlcul posterior. Per assignar un nom utilitzem uns **dos punts** seguit del **signe igual** (és a dir `nom := resultat`). En la línia que ve a continuació hem assignat a la lletra `k` el resultat del càlcul anterior. Després apliquem `evalf()` a `k`.

```
> k:=3/5+5/9+7/12;
> evalf(k);
```

Nota important de Maple: Maple distingeix entre majúscules i minúscules. Per tant Maple considera que `k` i `K` són variables diferents.

```
> k;
> K;
```

Per cert també es poden utilitzar paraules com a nom de variables.

```
> josep:=2^5;
> sqrt(josep);
```

Exemple 1.3

Si volem més o menys dígits de precisió en comptes dels 10 que s'utilitzen per defecte podem afegir un argument extra a la comanda `evalf()` tal i com es mostra a continuació.

```
> w:=4*(3+Pi);
> evalf(w);
> evalf(w,4);
> evalf(w,45);
```

Exemple 1.4

Si introduïu nombres amb un punt decimal Maple donarà de forma automàtica un resultat decimal. Compareu els dos resultats que apareixen a continuació.

```
> sqrt(34);
> sqrt(34.0);
```

Aquí hi ha un altre exemple:

```
> 4-1/3;
> 4.0-1/3;
```

Exemple 1.5

Podem aplicar la comanda `evalf()` a una successió de nombres: aquí sota generem primer les arrels quadrades exactes dels 10 primers nombres naturals, després apliquem `evalf()` per a obtenir les aproximacions decimals.

```
> result:=seq(sqrt(k),k=1..10);  
> evalf(result);
```

Drecera de Maple: referir-se de manera ràpida a l'últim resultat

Quan utilitzeu Maple hi ha moltes vegades que voldreu encadenar una sèrie de càlculs. En comptes de donar un nom a cada un dels resultats que aneu obtenint, podeu utilitzar el signe tant per cent (`%`) per referir-vos a l'última expressió que Maple ha calculat. Aquí hi ha alguns exemples de com funciona.

```
> 3/5+5/9+7/12;  
> evalf(%);  
> Pi;  
> evalf(%);  
> %+5;
```

Per a més informació de com fer servir aquest símbol mireu l'*Exercici 1.4* que ve tot seguit.

Exercici 1.1

Utilitzeu Maple per a calcular el nombre 37^{43} .

Exercici 1.2

Calculeu $\sqrt{34}$ amb 18 xifres.

Exercici 1.3

Obteniu una aproximació numèrica per a l'expressió : $\frac{3 + \pi}{7 - \sqrt{13}}$

Exercici 1.4

El símbol tant per cent (%) és una drecera molt útil però pot donar en alguns casos resultats inesperats. Aquí hi ha un exemple.

Primer executeu cada una de les tres línies següents. Heu de poder dir quin serà el resultat per endavant.

```
> 4+Pi;  
> evalf(%);  
> %+10;
```

Ara torneu enrera i torneu a executar l'última línia (i.e. > %+10;). Fixeu-vos que el resultat canvia de 17.14159265 a 27.14159265

Podeu explicar el perquè?

1.3 “Netejant” Variables

Des del moment en que definiu una variable, Maple recordarà quin és el seu valor durant tota la sessió de treball. Si voleu introduir un nou valor a la variable, només heu de fer la nova assignació.

Per exemple cada una de les assignacions que hi ha aquí baix redefineix el valor de la variable h. (Nota: per a verificar el valor actual d'una variable només cal escriure el seu nom seguit d'un punt i coma).

```
> h;  
> h:=56;  
> h;  
> h:=sqrt(Pi);  
> h;
```

A vegades voldreu “netejar” una variable de la memòria per a poder utilitzar-la en una situació nova. Aquí hi ha un exemple. Primer introduïm en x el valor 65.

```
> x:=65;
```

Ara suposem que comencem un problema nou i que volem introduir l'expressió algebraica genèrica $x^2 - 4x + 7$ i assignar-li el nom w. Si només fem això, Maple substitueix automàticament el valor anterior de x.

```
> w:=x^2-4*x+7;
```

Per a fer que x sigui una variable genèrica un altre cop hem de “netejar” (és a dir esborrar la memòria de Maple) el nostre antic valor per a x. Això s'aconsegueix introduint

```
> x := 'x';
```

Fixeu-vos que aquí utilitzem una cometa (apòstrof).

Executeu les dues línies següents per a veure com funciona.

```
> x:='x';  
> w:=x^2-4*x+7;
```

Netejant totes les variables de cop: la comanda restart.

La comanda `restart` neteja la memòria de Maple de totes les definicions que hagueu fet. És com si es comencés una nova sessió de Maple. Si heu de començar un problema totalment nou podeu utilitzar la comanda `restart` per assegurar que no queden definicions del vostre treball anterior. Abans d'executar la segona línia d'aquí sota, endevineu ràpidament el resultat.

```
> p:=4;  
> p; x; h;
```

Probablement recordareu que `p` era 4 i que `x` s'havia reassignat a la variable genèrica `x`, però poder no haureu recordat que `h` s'havia assignat al valor $\sqrt{\pi}$. És per això que és una bona idea utilitzar `restart` per a eliminar totes les definicions d'un sol cop. (Si aneu seguint aquestes pràctiques veureu que començarem moltes seccions noves amb una comanda `restart`).

```
> restart;  
> p; x; h;
```